Gauss principle pendulum

Holonomic constraints

Definition

Fixed distance to the point of fixation

In[1]:=
```
σ1 = x[t]^2 + (z[t] - l)^2 == l^2
```
Out[1]= $x[t]^2 + (-l + z[t])^2 == l^2$

Motion in the x-z plane

In[2]:=
```
σ2 = y[t] == 0
```
Out[2]= $y[t] == 0$

Resulting velocity constraints

In[3]:=
```
Vconstr1 = D[σ1, t]
```
Out[3]= $2\, x[t]\, x'[t] + 2\, (-l + z[t])\, z'[t] == 0$

In[4]:=
```
Vconstr2 = D[σ2, t]
```
Out[4]= $y'[t] == 0$

Resulting acceleration constraints

In[5]:=
```
Aconstr1 = D[σ1, {t, 2}]
```
Out[5]= $2\, x'[t]^2 + 2\, z'[t]^2 + 2\, x[t]\, x''[t] + 2\, (-l + z[t])\, z''[t] == 0$

Corresponding constraint function fulfilling $g_1[\ddot{\boldsymbol{r}}] \equiv 0$

In[6]:=
```
g1 = Aconstr1[[1]]
```
Out[6]= $2\, x'[t]^2 + 2\, z'[t]^2 + 2\, x[t]\, x''[t] + 2\, (-l + z[t])\, z''[t]$

In[7]:=
```
Aconstr2 = D[σ2, {t, 2}]
```
Out[7]= $y''[t] == 0$

Corresponding constraint function fulfilling $g_2[\ddot{\boldsymbol{r}}] \equiv 0$

In[8]:=
```
g2 = Aconstr2[[1]]
```
Out[8]= $y''[t]$

Q-function

Gravitational force

In[9]:=
```
F = -m * g * {0, 0, 1}
```

Out[9]= $\{0, 0, -g\,m\}$

Acceleration

In[10]:=
```
a = {x''[t], y''[t], z''[t]}
```

Out[10]= $\{x''[t], y''[t], z''[t]\}$

Extended Q function

In[11]:=
```
Qext = m / 2 (a - F / m) . (a - F / m) + μ1 * g1 + μ2 * g2
```

Out[11]= $$\mu2\, y''[t] + \mu1 \left(2\, x'[t]^2 + 2\, z'[t]^2 + 2\, x[t]\, x''[t] + 2\, (-l + z[t])\, z''[t]\right) + \frac{1}{2} m \left(x''[t]^2 + y''[t]^2 + (g + z''[t])^2\right)$$

Minimize Q-function

Gradient of Q with respect to the accelerations

In[12]:=
```
MatrixForm[GradQ = Grad[Qext, a]]
```

Out[12]//MatrixForm=
$$\begin{pmatrix} 2\,\mu1\, x[t] + m\, x''[t] \\ \mu2 + m\, y''[t] \\ 2\,\mu1\, (-l + z[t]) + m\, (g + z''[t]) \end{pmatrix}$$

General form of the accelerations

In[13]:=
```
asol = Table[Solve[GradQ[[i]] == 0, a[[i]]], {i, 1, 3}]
```

Out[13]= $$\left\{\left\{\left\{x''[t] \to -\frac{2\,\mu1\, x[t]}{m}\right\}\right\}, \left\{\left\{y''[t] \to -\frac{\mu2}{m}\right\}\right\}, \left\{\left\{z''[t] \to \frac{-g\,m + 2\, l\,\mu1 - 2\,\mu1\, z[t]}{m}\right\}\right\}\right\}$$

In[14]:=
```
aGen = Table[FullSimplify[asol[[i, 1, 1, 2]]], {i, 1, 3}]
```

Out[14]= $$\left\{-\frac{2\,\mu1\, x[t]}{m}, -\frac{\mu2}{m}, -\frac{g\,m - 2\, l\,\mu1 + 2\,\mu1\, z[t]}{m}\right\}$$

Compute Lagrange multipliers

In[15]:=
```
ReplacementRulesA = asol[[All, 1, 1]]
```

Out[15]= $$\left\{x''[t] \to -\frac{2\,\mu1\, x[t]}{m}, y''[t] \to -\frac{\mu2}{m}, z''[t] \to \frac{-g\,m + 2\, l\,\mu1 - 2\,\mu1\, z[t]}{m}\right\}$$

In[16]:=
```
Eq1 = (g1 == 0) /. ReplacementRulesA
```

Out[16]= $$-\frac{4\,\mu1\, x[t]^2}{m} + \frac{2\, (-l + z[t])\, (-g\,m + 2\, l\,\mu1 - 2\,\mu1\, z[t])}{m} + 2\, x'[t]^2 + 2\, z'[t]^2 == 0$$

In[17]:= `Eq2 = (g2 == 0) /. ReplacementRulesA`

Out[17]= $-\frac{\mu 2}{m} == 0$

In[18]:= `μsol = FullSimplify[Solve[{Eq1, Eq2}, {μ1, μ2}], Assumptions → {σ1, σ2, Vconstr1, Vconstr2}]`

Out[18]= $\left\{\left\{\mu 1 \to \frac{m\left(g\, l - g\, z[t] + x'[t]^2 + z'[t]^2\right)}{2\, l^2},\ \mu 2 \to 0\right\}\right\}$

Final equation of motion

In[19]:= `ReplacementRulesMu = μsol[[1, All]]`

Out[19]= $\left\{\mu 1 \to \frac{m\left(g\, l - g\, z[t] + x'[t]^2 + z'[t]^2\right)}{2\, l^2},\ \mu 2 \to 0\right\}$

In[20]:= `aGen`

Out[20]= $\left\{-\frac{2\,\mu 1\, x[t]}{m},\ -\frac{\mu 2}{m},\ -\frac{g\, m - 2\, l\, \mu 1 + 2\,\mu 1\, z[t]}{m}\right\}$

In[21]:= `aSol = FullSimplify[aGen /. ReplacementRulesMu, Assumptions → {σ1, σ2, Vconstr1, Vconstr2}]`

Out[21]= $\left\{-\frac{x[t]\left(g\, l - g\, z[t] + x'[t]^2 + z'[t]^2\right)}{l^2},\ 0,\ \frac{g\, z[t]^2 + l\left(x'[t]^2 + z'[t]^2\right) - z[t]\left(2\, g\, l + x'[t]^2 + z'[t]^2\right)}{l^2}\right\}$

In[22]:= `EqMotGauss = Table[a[[i]] == aSol[[i]], {i, 1, 3}];`

This equation of motion is equivalent to the one obtained by D’Alembert’s principle

In[23]:= `MatrixForm[EqMotGauss]`

Out[23]//MatrixForm=

$$\begin{pmatrix} x''[t] == -\frac{x[t]\left(g\, l - g\, z[t] + x'[t]^2 + z'[t]^2\right)}{l^2} \\ y''[t] == 0 \\ z''[t] == \frac{g\, z[t]^2 + l\left(x'[t]^2 + z'[t]^2\right) - z[t]\left(2\, g\, l + x'[t]^2 + z'[t]^2\right)}{l^2} \end{pmatrix}$$